

Supplementary Code 1: WinBUGS code for the BaySTDetect model

```
model {
  for (i in 1:N) {
    for (t in 1:T) {
      y[i,t] ~ dpois(E.mu[i,t])
      E.mu[i,t] <- E[i,t] * mu[i,t]
      log(mu[i,t]) <- p[i] * crw.mix[i,t] + (1-p[i]) * rw.mix[i,t]
    }
    p[i] ~ dbern(0.95)
  }

# Common trend model
for (i in 1:N) {
  for (t in 1:T) {
    y1[i,t] ~ dpois(mu1[i,t])
    log(mu1[i,t]) <- log(E[i,t]) + temp1[i,t]
    temp1[i,t] <- alpha + eta[i] + gamma[t] + beta * x[i,t]
    crw.mix[i,t] <- cut(temp1[i,t])
  }
}
alpha ~ dnorm(0,0.001)
gamma[1:T] ~ car.normal(adj.t[],weights.t[],num.t[],tau.gamma)
beta ~ dnorm(0, 0.001)
eta[1:(N-1)] ~ car.normal(adj[],weights[],num[],tau.eta)
eta[N] <- 0 # SLA with zero neighbours
tau.gamma <- pow(sigma.gamma,-2)
sigma.gamma ~ dnorm(0,1)I(0,)
tau.eta <- pow(sigma.eta,-2)
sigma.eta ~ dnorm(0,1)I(0,)

# Area-specific Model
for (i in 1:N) {
  for (t in 1:T) {
    y2[i,t] ~ dpois(mu2[i,t])
    log(mu2[i,t]) <- log(E[i,t]) + temp2[i,t]
    temp2[i,t] <- u[i] + xi[i,t] + beta.dash * x[i,t]
    rw.mix[i,t] <- cut(temp2[i,t])
  }
  u[i] ~ dnorm(0,0.001)
  xi[i,1:T] ~ car.normal(adj.t[],weights.t[],num.t[],tau.xi[i])
  tau.xi[i] <- pow(var.xi[i],-1)
  var.xi[i] <- exp(log.var.xi[i])
  log.var.xi[i] ~ dnorm(a,cc)
}
beta.dash ~ dnorm(0, 0.001)
a ~ dnorm(0,0.001)
cc <- pow(c,-2)
c ~ dnorm(0,d)I(0,)
d <- pow(2.5,-2)
}
```